17.2.2 POINT CHARGE IN FRONT OF A GROUNDED CONDUCTING SPHERE

B. Schnizer

schnizer@itp.tu-graz.ac.at

2013-01-21

Summary:
A point charge e is at a distance r' from the centre of a ideally conducting grounded sphere of radius a. The static potential is computed by the method of image charges. The position and the value of the image charge are found by inversion, i.e. by reflection on the sphere. From this are computed both the surface charge distribution on the sphere and the electric field strength. The force exerted on the sphere by the point charge is computed in two ways:

1. The force of the electric field on the charge density is integrated over the whole surface.
2. The Coulomb force between the real charge e and the image charge is computed.

The integral over the surface charge density gives the value of the image charge.
The two results for the force differ by a factor 1/2 ! The correct result is that to method 2. Method 1 includes the self-interaction of a surface element with itself, which does **not** belong force exerted by the point charge.
There is a general proof that this spurious interaction is just the half ot the total interaction.

Data

a radius of sphere
rp radial distance of point charge e from centre of sphere
thp = 0 ⇒ problem is axially symmetrical.
r, th coordinates of field point

Auxiliary Commands

- **Transformation from spherical to cylindrical coordinates**

```
sucy = {r^2 → ρ^2 + z^2, c → z / r}
```

$$\left\{r^2 \to z^2 + \rho^2,\ c \to \frac{z}{r}\right\}$$

- **Transformation from cylindrical to spherical coordinates**

```
susp = { ρ → r Sin[th], z → r Cos[th]}
```

```
{ρ → r Sin[th], z → r Cos[th]}
```

- **Abbreviation c is used for cos(th)**

```
suc = {c → Cos[th]}
```

```
{c → Cos[th]}
```

- **surface element on sphere integrated over azimuth**

```
df = 2 π a^2 Sin[th]
```

$2\,a^2\,\pi\,\mathrm{Sin}[th]$

Potential, SK (17.26), and electric field

```
Φ = e (4 π ε Sqrt[r^2 - 2 r c rp + rp^2])^(-1) -
  e (4 π ε Sqrt[a^2 - 2 r c rp + rp^2 r^2 / a^2])^(-1)
```

$$\frac{e}{4\,\pi\,\sqrt{r^2 - 2\,c\,r\,rp + rp^2}\;\varepsilon} - \frac{e}{4\,\pi\,\sqrt{a^2 - 2\,c\,r\,rp + \frac{r^2\,rp^2}{a^2}}\;\varepsilon}$$

- **Potential in cylindrical coordinates**

```
Φz = Φ /. sucy
```

```
General::spell1 : Possible spelling error:
   new symbol name "Φz" is similar to existing symbol "Φ". More…
```

$$\frac{e}{4\,\pi\,\varepsilon\,\sqrt{rp^2 - 2\,rp\,z + z^2 + \rho^2}} - \frac{e}{4\,\pi\,\varepsilon\,\sqrt{a^2 - 2\,rp\,z + \frac{rp^2\,(z^2+\rho^2)}{a^2}}}$$

- ***E_z***

```
ez = - D[Φz, z]
```

$$\frac{e\,(-2\,rp + 2\,z)}{8\,\pi\,\varepsilon\,\left(rp^2 - 2\,rp\,z + z^2 + \rho^2\right)^{3/2}} - \frac{e\,\left(-2\,rp + \frac{2\,rp^2\,z}{a^2}\right)}{8\,\pi\,\varepsilon\,\left(a^2 - 2\,rp\,z + \frac{rp^2\,(z^2+\rho^2)}{a^2}\right)^{3/2}}$$

- **E_z depending on spherical coordinates**

```
ez /. susp
```

$$\frac{e\,(-2\,rp + 2\,r\,\mathrm{Cos}[th])}{8\,\pi\,\varepsilon\,\left(rp^2 - 2\,r\,rp\,\mathrm{Cos}[th] + r^2\,\mathrm{Cos}[th]^2 + r^2\,\mathrm{Sin}[th]^2\right)^{3/2}} - \frac{e\,\left(-2\,rp + \frac{2\,r\,rp^2\,\mathrm{Cos}[th]}{a^2}\right)}{8\,\pi\,\varepsilon\,\left(a^2 - 2\,r\,rp\,\mathrm{Cos}[th] + \frac{rp^2\,(r^2\,\mathrm{Cos}[th]^2 + r^2\,\mathrm{Sin}[th]^2)}{a^2}\right)^{3/2}}$$

- E_z **on sphere**

```
eza = % /. r → a
```

$$\frac{e\,(-2\,rp + 2\,a\,Cos[th])}{8\,\pi\,\varepsilon\,\left(rp^2 - 2\,a\,rp\,Cos[th] + a^2\,Cos[th]^2 + a^2\,Sin[th]^2\right)^{3/2}} - \frac{e\left(-2\,rp + \frac{2\,rp^2\,Cos[th]}{a}\right)}{8\,\pi\,\varepsilon\left(a^2 - 2\,a\,rp\,Cos[th] + \frac{rp^2\,(a^2\,Cos[th]^2 + a^2\,Sin[th]^2)}{a^2}\right)^{3/2}}$$

Surface charge density on sphere, SK(after (17.26))

- E_r

```
er = - D[Φ, r]
```

$$\frac{e\,(2\,r - 2\,c\,rp)}{8\,\pi\,\left(r^2 - 2\,c\,r\,rp + rp^2\right)^{3/2}\,\varepsilon} - \frac{e\left(-2\,c\,rp + \frac{2\,r\,rp^2}{a^2}\right)}{8\,\pi\left(a^2 - 2\,c\,r\,rp + \frac{r^2\,rp^2}{a^2}\right)^{3/2}\,\varepsilon}$$

- $\eta = \varepsilon\ E_r$ **at r = a**

```
et = ε er /. r → a // Together
```

$$\frac{a^2\,e - e\,rp^2}{4\,a\,\pi\,\left(a^2 - 2\,a\,c\,rp + rp^2\right)^{3/2}}$$

```
etsc = % /. suc
```

$$\frac{a^2\,e - e\,rp^2}{4\,a\,\pi\,\left(a^2 + rp^2 - 2\,a\,rp\,Cos[th]\right)^{3/2}}$$

- **Total charge on sphere**

```
Integrate[df etsc, {th, 0, π}, Assumptions → a > 0 && rp > a]
```

$$-\frac{a\,e}{rp}$$

- e a / r' = - e r'' / a = e''

Total force on sphere

- **dFz**

- ***E_z* at r = a**

```
eza
```

$$\frac{e\,(-2\,rp+2\,a\,\text{Cos}[th])}{8\,\pi\,\varepsilon\,\left(rp^2-2\,a\,rp\,\text{Cos}[th]+a^2\,\text{Cos}[th]^2+a^2\,\text{Sin}[th]^2\right)^{3/2}}-\frac{e\,\left(-2\,rp+\frac{2\,rp^2\,\text{Cos}[th]}{a}\right)}{8\,\pi\,\varepsilon\,\left(a^2-2\,a\,rp\,\text{Cos}[th]+\frac{rp^2\,\left(a^2\,\text{Cos}[th]^2+a^2\,\text{Sin}[th]^2\right)}{a^2}\right)^{3/2}}$$

- ***E_z* at r = a times charge density times surface element df**

```
eza etsc df
```

$$\left(a\,\left(a^2\,e-e\,rp^2\right)\,\text{Sin}[th]\,\left(\frac{e\,(-2\,rp+2\,a\,\text{Cos}[th])}{8\,\pi\,\varepsilon\,\left(rp^2-2\,a\,rp\,\text{Cos}[th]+a^2\,\text{Cos}[th]^2+a^2\,\text{Sin}[th]^2\right)^{3/2}}-\frac{e\,\left(-2\,rp+\frac{2\,rp^2\,\text{Cos}[th]}{a}\right)}{8\,\pi\,\varepsilon\,\left(a^2-2\,a\,rp\,\text{Cos}[th]+\frac{rp^2\,\left(a^2\,\text{Cos}[th]^2+a^2\,\text{Sin}[th]^2\right)}{a^2}\right)^{3/2}}\right)\right)\Big/\left(2\,\left(a^2+rp^2-2\,a\,rp\,\text{Cos}[th]\right)^{3/2}\right)$$

- **Fz = total fource**

```
Integrate[%, {th, 0, π}, Assumptions → a > 0 && rp > a]
```

$$\frac{a\,e^2\,rp}{2\,\pi\,\left(a^2-rp^2\right)^2\,\varepsilon}$$

This result is wrong ! It includes the self-interaction of a surface element with itself, which does **not** belong to the force exerted by the point charge. There is a general proof that this spurious interaction is just the half ot the total interaction. The correct result is derived just below.

- **Coulomb force between point charge and image point charge:**

F = e e'' /(4 π ε)/(r' - r'')^2 = - $e^2\,a/r'$ /(4 $\pi\,\varepsilon$) / (r' - a^2/r')^2 =

= - $e^2/(4\,\pi\,\varepsilon)$ a r' /($r'^2 - a^2$)^2 = - e^2 a r' / ((4 π ε) ($r'^2 - a^2$)^2)

```
F = -e^2 a rp / ((4 π ε) (rp^2 - a^2)^2)
```

$$\frac{-a\,e^2\,rp}{4\,\pi\,\left(a^2-rp^2\right)^2\,\varepsilon}$$

This is the correct result.